# $Id: PKGBUILD 166514 2012-09-09 07:34:43Z allan $
# Maintainer: Allan McRae <allan@archlinux.org>

pkgname=libmpc
pkgver=1.0.1
pkgrel=1
pkgdesc="Library for the arithmetic of complex numbers with arbitrarily high precision"
arch=('i686' 'x86_64')
url="http://www.multiprecision.org/"
license=('LGPL')
depends=('mpfr>=3.0.0')
options=('!libtool')
install=libmpc.install
source=(http://www.multiprecision.org/mpc/download/mpc-${pkgver/_/-}.tar.gz{,.sig})
md5sums=('b32a2e1a3daa392372fbd586d1ed3679'
         'c064863fbbb5831ba4f75f1181333862')

build() {
  cd "${srcdir}/mpc-${pkgver}"
  ./configure --prefix=/usr
  make
}

check() {
  cd "${srcdir}/mpc-${pkgver}"
  make check
}

package() {
  cd "${srcdir}/mpc-${pkgver}"
  make DESTDIR="${pkgdir}" install
  mv ${pkgdir}/usr/share/info/{mpc,libmpc}.info
}
